PSpice Simulation

• The target of computer-aided analysis is to determine the circuit currents and voltages everywhere in the circuit.

• For PSpice, the circuit is described by a text file called the netlist.

• Three types of statements are used in this netlist: circuit description statements, simulation commands and program control statements.

• The netlist may be entered directly using a text editor or a schematic capture program with a graphical user interface automatically generates the netlist.

• References

M.H. Rashid, Introduction to PSpice Using OrCad for Circuits and Electronics, Third Edition

B.M. Wilamowski and R.C. Jaeger, Computerized circuit analysis using SPICE programs

R.R. Spencer and M.S. Ghausi, Introduction to Electronic Circuit Analysis, Chapter 4

Downloads

PSpice with Schematic capture

http://www.orcad.com/downloads/demo/default.asp
Download PSpice 9.1 Student Version

WinSpice3

http://www.willingham2.freeserve.co.uk/winspice.html

and select download Spice3F4

There are virtually hundreds of web links that will allow you to download different versions of PSpice

A Low Pass Filter




```
Low-Pass Filter
*circuit description statements
C1          out 0  1u
R2          out 0  1k
R1          in out  1k
vin        in 0 DC 0Vdc AC 1Vac
* simulation commands
.probe
.ac dec 10 .1 1k
.end
```

The results of the simulation are stored in a .out file for later plotting.

A Low Pass Filter (Cont'd)



Simulation results

Things to Remember

- Every two terminal component will have a positive node and a negative node.

- The direction of the +ve current is from the +ve node to the –ve node through the component.

- All possible currents and voltages can be stored/plotted in PSpice. They do, however, have a special format. Examples are:

v(5)	voltage at node 5 with respect to ground
v(4,2)	voltage at node 4 with respect to node 2
v(R1)	voltage of resistor R1 (polarity observed!)
v(L1)	voltage of inductor L1 (polarity observed!)
v(C1)	voltage of capacitor C1(Polarity observed!)
I(Vs)	current through the source Vs
I(R5)	current through the resistor R5

Simulation Types in PSpice

- There are three simulation types in PSpice: DC, AC and Transient

- In the DC mode, only DC sources are present. Capacitors are replaced by open circuit and inductors are replaced by short circuit. PSpice solves for the voltages and currents iteratively.

- In the AC mode, only AC sources are present. All components are replaced by their complex impedances. PSpice solves for the steady state sinusoidal currents and voltages through a modified nodal analysis.

- In the transient mode, the sources may take any arbitrary waveform in time. Capacitors and inductors are replaced by their associated differential equations. PSpice then obtains the voltages and currents everywhere in the circuit at each instant of time through an iterative approach.

- There are different possible simulation analyses within each simulation type.

DC Simulation Analyses



.op directive obtains the operating point of the circuit

*Analysis directives:
.OP
.PROBE V(*) I(*) W(*) D(*) NOISE(*)
C_C1 0 N00037 1n
L_L1 N00239 N00475 10uH
R_R2 0 N00037 1k
R_R1 N00475 N00037 1k
V_V1 N00239 0 5v
R_R3 0 N00475 1k
.END

The output file contains the following results:

NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE
(N00037) 2.5000 (N00239) 5.0000 (N00475) 5.0000

VOLTAGE SOURCE CURRENTS
NAME CURRENT

V_V1 -7.500E-03

DC Simulation Analyses (Cont'd)



.TF directive obtains the transfer function between two sets of input pairs

VIN 1 0 DC 1V
R1 1 2 1k
R2 2 0 20k
Rp 2 6 1.5k
RE 3 0 250
F1 4 3 Vx 40 ; current controlled current source
R0 4 3 100k
RC 4 3 2k
Vx 6 3 DC 0v
Vy 5 0 DC 0v
.TF V(4) VIN
.END

the output file includes the following statements

**** SMALL-SIGNAL CHARACTERISTICS

V(4)/Vin = -5.969E+00
INPUT RESISTANCE AT Vin = 8.313E+03
OUTPUT RESISTANCE AT V(4) = 1.992E+03

DC Simulation Analyses (Cont'd)



.DC directive allows us to sweep sources and parameters

```
Bridge circuit
.probe
.dc vin  4.9  5.1  .001
vin  1  0  dc  5
v_ir5  3  4  dc  0v
r1  2  3  1k
r2  1  2  1k
r3  2  0  1k
r4  1  3  2k
r5  4  0  3k
.end
```



<u>AC Simulation Analyses</u>



.AC directive enables a frequency sweep of the frequency

Resonant Circuit
Vin 1 0 ac 1m
L1 1 2 3.3m
C1 2 3 10n
Rs 3 0 2.2
.ac dec 100 1k   100k
.print ac V(3)
.End



Other types of AC analysis are .Noise, .Disto and .pz. They carry out noise analysis, distortion analysis and pole-zero analysis of the circuit.

Transient Simulation Analyses



* PULSE(-Vs Vs TDelay TRise TFall PulseWidth Period)
Vin 1 0 PULSE(-220V 220V 0 1ns 1ns 100us 200us)
L1 1 2 50u
C1 2 3 10u
Rs 3 0 2.2
.TRAN 1US 400US
.Probe
.END



Transient Simulation Analyses




```
* PWL (T1  V1   T2   V2   T3   V3  . . .)
Vin1  1   0  PWL (0   0   1NS    1V     1ms    1V)
Vin2  4   0  PWL (0   0   1NS    1V     1ms    1V)
Vin3  7   0  PWL (0   0   1NS    1V     1ms    1V)
R1    1    2   2
L1    2    3   50u
C1    3    0    10u
R2    4    5    1
L2    5    6    50u
C2    6    0    10u
R3    7    8      8
L3    8    9     50u
C3    9    0     10u
.TRAN   1us   400us
.PROBE
.END
```



Transient Simulation Analyses




```
* sin(Voffset, amplitude, frequency)
Vin   in   0   sin (0  20   60)
D1  in    out  D1N4002
R1   out    0   1k
C1   out   0   100u
.model   D1N4002   D(IS   14.110000E-09, N    1.984,ISR  100.000000E-12,
 +IKF   94.81, BV  100.1,  IBV   10, RS     .03389, TT    4.761000E-06,
+CJO   51.170000E-12, VJ     .3905, M     .2762)
.end
```



Two Terminal Elements Symbols

C Capacitor D Diode
E voltage-controlled voltage source
F Current-controlled current source
G Voltage-controlled current source
H Current-controlled voltage source
I Independent current source
L Inductor R resistor
V Independent voltage source
T Transmission Line

Three or Four Terminal Devices

B GaAs MESFET (D=Drain, G=Gate, S=Source)
J JFET (D=Drain, G=Gate, S=Source)
M MOSFET (D=Drain, G=Gate, S=Source, B=Bulk)
Q BJT (C=Collector, B=Base, E=Emitter, S=Substrate)